

## A Combined AE + CM Graph using SAS

Sanjay Matange, SAS Institute Inc.

### ABSTRACT

Patient profile graphs generally include visual display of the clinical data for one subject on a timeline. The graphs include display of the Adverse Events by severity, Concomitant Medications and other graphs such as Vital Statistics, Lab Results and other plots over time.

This paper will demonstrate how to combine some of these graphs into one graph, where the Adverse Events and the Concomitant Medications can be displayed in the same graph, interspersed by the start date for each event or medication on a common time line. Such a graph places related information in close proximity thus improving the understanding of the data. This technique can also be extended to include more data in one graph.

### INTRODUCTION

The SAS SG Procedures provide you with powerful features to create graphs commonly used in the Health and Life Sciences domain. The SGPLOT procedure creates single-cell graphs and supports many plot and supporting statements. Many graphs can be created by use of a single plot statement and most graphs can be created by layering compatible plot statements. One such versatile plot statement introduced with SAS 9.3 is the HIGHLOW plot statement. This statement was designed specifically to simplify the building of Adverse Events Timeline plots as we will see in this presentation.

A Patient Profile report often includes multiple graphs that display the data for one subject over a timeline. These reports often include separate display of the adverse events by severity and concomitant medications over time. Often such graphs are created separately and care must be taken to ensure the x-axis timelines for both graphs is made uniform to allow easier understanding of the data.

In this presentation, I will show you how you can combine the AE and CM data into a common SAS data set so data from both domains can be displayed in one graph with a common timeline. Placing these data in one graph using one common x-axis aligns the events correctly and having them in close proximity improves the understanding of the data and any interactions.

### THE ADVERSE EVENT TIMELINE GRAPH

For this presentation, I have used simulated AE and CM data. Often, some cleaning of the data may be required while converting it into the form suitable for plotting using the SGPLOT procedure. Let us start with the AE plot. The data for the AE graph is shown in Figure 1.

aestdtc	aeseq	aedecod	aesev	aeout	aeendtc	aestdy	aeendy
2013-03-06	1	DIZZINESS	MODERATE	RESOLVED	2013-03-06	3	5
2013-03-20	2	COUGH	MILD	NOT RESOLVED		17	.
2013-03-26	3	APPLICATION SITE DERMATITIS	MODERATE	NOT RESOLVED	2013-06-18	23	107
2013-03-27	4	DIZZINESS	MILD	RESOLVED	2013-03-27	24	25
2013-03-30	5	ELECTROCARDIOGRAM T WAVE INVERSION	MILD	NOT RESOLVED		27	.
2013-04-01	6	DIZZINESS	MILD	NOT RESOLVED	2013-04-11	29	39
2013-05-17	9	HEADACHE	MILD	RESOLVED	2013-05-18	75	78
2013-05-27	11	PRURITUS	MODERATE	RESOLVED	2013-06-18	85	107

Figure 1 – AE data set.

The columns used to create this graph are shown in Figure 1. The observations in the data set represent all the adverse events for the subject. Each observation represents one adverse event, with a start date

character string, sequence number, the adverse event, severity, the outcome, end date character string, a start day and end day.

In this data, some events do not have an end date, or a start or end day. There may be some repetitions in the data. This needs to be cleaned. In our case, I find the lowest start day (minday) and the highest end day (maxday). Then, I compute a new data set as follows.

1. If aestdy is missing, set it to min day and set a new column LOCAP to FilledArrow.
2. If aeendy is missing, set it to max day and set a new column HICAP to FilledArrow.
3. Find the lowest start date, and compute all other dates from this and aestdy.

Here is the new derived data set with the new columns. The NAME column contains the aedecod values and the LABEL column contains the name for the first occurrence of the aedecod. LoCap and HiCap columns contain the text FILLEDARROW when the start or end date are missing in the original data set.

Obs	startdate	enddate	sev	stdy	endy	name	LoCap	HiCap	label
1	2013-03-06	2013-03-08	Moderate	3	5	DIZZINESS			Dizziness
2	2013-03-20	2013-06-18	Mild	17	107	COUGH		FILLEDARROW	Cough
3	2013-03-26	2013-06-18	Moderate	23	107	APPLICATION SITE DERMATITIS			Application Site Dermatitis
4	2013-03-27	2013-03-28	Mild	24	25	DIZZINESS			
5	2013-03-30	2013-06-18	Mild	27	107	ELECTROCARDIOGRAM T WAVE INVERSION		FILLEDARROW	Electrocardiogram T Wave Inversion
6	2013-04-01	2013-04-11	Mild	29	39	DIZZINESS			
7	2013-05-17	2013-05-20	Mild	75	78	HEADACHE			Headache
8	2013-05-27	2013-06-18	Moderate	85	107	PRURITUS			Pruritus

**Figure 2 – Data set derived for plotting the AE graph**

The SGPLOT code for creating this graph is shown below.


```

title "Adverse Event for Patient Id = xx-xxx-xxxx";
proc sgplot data=ae3 dattrmap=attrmap;
  format stdate date7.;
  refline 0 / axis=x lineattrs=(color=black);
  highlow y=name low=stdy high=endy / type=bar group=sev
    lineattrs=(color=black pattern=solid)
    barwidth=0.8 lowlabel=label lowcap=locap highcap=hicap
    attrid=Severity nomissinggroup
    labelattrs=(size=8) dataskin=preserved;
  scatter y=name x=stdate / x2axis markerattrs=(size=0);
  xaxis grid display=(nolabel) values=(&minday to &maxday by 2);
  x2axis display=(nolabel) values=(&mindate to &maxdate);
  yaxis grid display=(noticks novalues nolabel);
run;

```

For the code shown above, the macro variables &minday, &maxday, &mindate and &maxdate are computed in the data step. &mindate and &maxdate are the corresponding dates for &minday and &maxday.

The resulting Adverse Event Timeline graph is shown in Figure 3.



**Figure 3 – Adverse Event timeline graph**

In the code above, we have used DATTRMAP=attrmap, where attrmap is a SAS data set that defines the visual attributes to be used to map the group colors for the high low statement with ATTRID=Severity. A visual of the data set is shown below.

Obs	id	value	fillcolor	linecolor	show
1	Severity	Mild	lightgreen	black	attrmap
2	Severity	Moderate	gold	black	attrmap
3	Severity	Severe	red	black	attrmap

**Figure 4 – Discrete Attributes Map**

Note in the graph shown in Figure 3, the discrete legend shows three levels for Severity, Mild, Moderate and Severe. The values in the graph only include Mild and Moderate. There are no Severe events in the data. However, in the discrete attributes map, we have defined all three values, and the “Show” column is set to “AttrMap”. This causes all the values from the attribute map to be displayed in the legend. It also useful to note that the values in the legend are sorted in the order of the values in the data set. So, this provides a convenient way to order the values in the legend.

### THE CONCOMITANT MEDICATIONS TIMELINE GRAPH

Now, let us create the Concomitant Medications Timeline graph. For this I use a simulated CM data set as shown in Figure 5. For our purpose, we will only use cmtrt, cmstdtc, cmstdy, cmendy, cmdtc variables. Some of the other variables can be used to construct a more detailed name for the medication displayed in the graphs.

cmtrt	cmstdtc	cmseq	visitdy	cmdose	cmdosu	cmdosfrq	cmstdy	cmendy	cmdtc
VICKS FORMULA 44D	2013-03-20	18	28	1	Teaspoon	ONCE A DAY	17	25	2013-03-30
HYDROCORTISONE	2013-03-27	19	28	1	Unit	AS NEEDED	24	107	2013-03-30
BECONASE	2013-03-30	20	28	1	SPRAY	2 TIMES A DAY	27	.	2013-03-30
IBUPROFEN	2013-05-17	41	84	500	Milligram	ONCE A DAY	75	78	2013-05-25

**Figure 5 – Concomitant Medications data**

As we did for the AE plot, the data is cleaned and the missing values are inserted including columns for drawing low and high caps. Note we have used the variable names that are consistent with the names in the data set used for the AE graph. We also added a Severity column with the values "Medication". The reason for this will be clear in the next section. The processed data is as shown in Figure 6.

Obs	stdate	enddate	sev	stdy	endy	name	LoCap	HiCap	label
1	2013-03-20	2013-03-28	Medication	17	25	VICKS FORMULA 44D			Vicks Formula 44d
2	2013-03-27	2013-06-18	Medication	24	107	HYDROCORTISONE			Hydrocortisone
3	2013-03-30	2013-06-18	Medication	27	107	BECONASE		FILLEDARROW	Beconase
4	2013-05-17	2013-05-20	Medication	75	78	IBUPROFEN			Ibuprofen

**Figure 6 - Data set derived for plotting the CM graph**

The SGPLOT code for creating this graph is shown below.


```

title "Medications for Patient Id = &pid";
proc sgplot data=cm3 dattrmap=attrmap noautolegend;
  format stdate date7.;
  refline 0 / axis=x lineattrs=(color=black);
  highlow y=name low=stdy high=endy / type=bar group=sev
    lineattrs=(color=black pattern=solid)
    barwidth=0.8 lowlabel=label lowcap=locap highcap=hicap
    attrid=Severity nomissinggroup
    labelattrs=(size=8) dataskin=pressed;
  scatter y=name x=stdate / x2axis markerattrs=(size=0);
  xaxis grid display=(nolabel) values=(&mindaym to &maxdaym by 2);
  x2axis display=(nolabel) values=(&mindatem to &maxdatem);
  yaxis grid display=(noticks novalues nolabel);
run;

```

In the code shown above, the macro variables &mindaym, &maxdaym, &mindatem and &maxdatem are computed in the data step for the medications data. &mindatem and &maxdatem are the corresponding dates for &mindaym and &maxdaym.

The resulting Concomitant Medications Timeline graph is shown in Figure 7.



**Figure 7 – Concomitant Medications timeline graph**

The graph above uses its own macros for the min and max day and data extents, so the x-axis spans from 17 to 107, which is not the same as the x-axis of the graph in Figure 3. We can compute the min and max values over both the AE and CM data sets and use macro variables that span the union on the data. Setting those values gives the CM graph shown in Figure 8. The AE graph from Figure 3 is repeated here for comparison.




Figure 8 – CM Graph with union data extent for x-axis




Figure 9 – Repeat of AE Graph for comparison

Even though both the graphs in Figure 8 and Figure 9 use the same axis extents, we can see the events in the graphs are not aligned. This is due to the fact that the AE names for in Figure 9 are longer and thus need a larger offset to be properly included in the graph without clipping. So, even though the axis extents in both graphs are 3 – 105 days, the events are not aligned, making it harder to understand the data.

### THE COMBINED AE + CM TIMELINE GRAPH

The correct solution here is to combine the events in both the AE and CM domains and plot a combined AE + CM graph as shown in Figure 10. This is easily done since we have converted both the AE data and the CM data into similar data sets with same column names. Now can simply append the modified CM data with the modified AE data to create a common data set with all the events in one place as shown in Figure 10.

Obs	stdate	endate	sev	stdy	endy	name	LoCap	HiCap	label
1	2013-03-06	2013-03-08	Moderate	3	5	DIZZINESS			Dizziness
2	2013-03-20	2013-06-18	Mild	17	107	COUGH		FILLEDARROW	Cough
3	2013-03-20	2013-03-28	Medication	17	25	VICKS FORMULA 44D			Vicks Formula 44d
4	2013-03-26	2013-06-18	Moderate	23	107	APPLICATION SITE DERMATITIS			Application Site Dermatitis
5	2013-03-27	2013-03-28	Mild	24	25	DIZZINESS			
6	2013-03-27	2013-06-18	Medication	24	107	HYDROCORTISONE			Hydrocortisone
7	2013-03-30	2013-06-18	Mild	27	107	ELECTROCARDIOGRAM T WAVE INVERSION		FILLEDARROW	Electrocardiogram T Wave Inversion
8	2013-03-30	2013-06-18	Medication	27	107	BECONASE		FILLEDARROW	Beconase
9	2013-04-01	2013-04-11	Mild	29	39	DIZZINESS			
10	2013-05-17	2013-05-20	Mild	75	78	HEADACHE			Headache
11	2013-05-17	2013-05-20	Medication	75	78	IBUPROFEN			Ibuprofen
12	2013-05-27	2013-06-18	Moderate	85	107	PRURITUS			Pruritus


**Figure 10 – Combined AE + CM data**

We can sort the data by stdy and plot the data with the code shown below to create the combined AE+CM graph.

```

title "AE + CM for Patient Id = &pid";
proc sgplot data=aecm datrmap=attrmapc;
  format stdate date7.;
  refline 0 / axis=x lineattrs=(color=black);
  highlow y=name low=stdy high=endy / type=bar group=sev
    lineattrs=(color=black pattern=solid)
    barwidth=0.8 lowlabel=label lowcap=locap highcap=hicap
    attrid=Severity nomissinggroup
    labelattrs=(size=7) dataskin=preserved;
  scatter y=name x=stdate / x2axis markerattrs=(size=0);
  xaxis grid display=(nolabel) values=(&mindayc to &maxdayc by 2);
  x2axis display=(nolabel) values=(&mindatec to &maxdatec);
  yaxis grid display=(noticks novalues nolabel);
run;

```



**Figure 11 – Combined AE + CM graph**

For this graph, we have updated the Discrete Attributes Map data set to include information for Medications. This allows us to display the three different severity values as green yellow and red, and the Medications values in blue.

Obs	id	value	fillcolor	linecolor	show
1	Severity	Mild	lightgreen	black	attrmap
2	Severity	Moderate	gold	black	attrmap
3	Severity	Severe	red	black	attrmap
4	Severity	Medication	lightblue	blue	attrmap

**Figure 12 – Combined Attributes Map**

The graph in Figure 11 aligns all the adverse events and medications correctly, with the right axis extents and offsets. Now it is easier to understand the data and any interactions between the events and the medications. This also conforms to the “Proximity” principle for effective graphics that recommends that values that need to be compared should be brought as close as possible.

## CONCLUSION

Graphs showing Adverse Events, Concomitant Medications and other information like dosage could be plotted over the same data range in separate graphs using the domain data. However, events in such graphs could still be misaligned due to details such as event or medication names, etc. This makes it harder to compare and understand the interactions between different graphs.

Plotting all relevant data in one combined graph has many advantages. This allows us to plot each adverse event, dosage or concomitant medications side by side correctly aligned with the same axis extents and offsets. This also brings the relevant information closer and thus makes the information and interactions between events easier to understand. The techniques demonstrated in this presentation can be used to bring more information into one graphs for better understanding of the information.

## RECOMMENDED READING

- Matange, Sanjay. SAS Global Forum 2013 “Patient Profile Graphs using SAS.”
- Matange, Sanjay. Graphically Speaking Blog: <http://blogs.sas.com/content/graphicallyspeaking/>
- Matange, S. 2016. Clinical Graphs using SAS. 1st ed. Cary, NC: SAS Press.

## CONTACT INFORMATION

Your comments and questions are valued and encouraged. Contact the author at:

Sanjay Matange  
 SAS Institute Inc.,  
 100 SAS Campus Dr.  
 Cary, NC 27513  
 Sanjay.Matange@sas.com  
<http://blogs.sas.com/content/graphicallyspeaking>

SAS and all other SAS Institute Inc. product or service names are registered trademarks or trademarks of SAS Institute Inc. in the USA and other countries. ® indicates USA registration.

Other brand and product names are trademarks of their respective companies.